
1 Introduction

The development of critical software has put forward
architecture description languages. The aim of these
languages is to allow a veri�cation process at the early
stages of software development. They cover many
aspects like real time, information �ow, distribution,
execution platforms, ...
The Architecture Analysis and Design Language

(AADL)[8] standard was prepared by the SAE
AS-2C Architecture Description Language Subcom-
mittee, Embedded Computing Systems Committee,
Aerospace Avionics Systems Division. The AADL
standard is based on MetaH[9], an architecture de-
scription language developed at Honeywell Labora-
tories under the sponsorship of the US Defense Ad-
vanced Research Projects Agency (DARPA) and US
Army Aviation and Missile Command (AMCOM).
Release 1.0 of the AADL standard (SAE AS5506)
has been issued in November 2004.
The AADL is a language used to describe the

software and hardware components of a system and
the interfaces between those components. The lan-
guage can describe functional interfaces of compo-
nents, such as data inputs and outputs, and non-
functional aspects, such as timing properties. The
language can describe how components are combined,
such as how data inputs and outputs are connected
or how software components are allocated to hard-
ware components. More detailed information about
this language may be found at: www.aadl.info.
Although some behavioural aspects can be de-

scribed in AADL (e.g. static linear control �ows),
application behaviours rely mainly on source code
written in a target language (C, Ada, ...). This pa-
per presents a behaviour extension as properties and
a dedicated annex in order to make possible a better
analysis of behaviours. These extensions try to reuse
as much as possible existing AADL concepts. The
main features of our proposal concern:

• High level composition paradigms: we have stud-
ied how to introduce HRT-HOOD [4] composi-
tion paradigms within AADL.

• Dynamic aspects related to subprogram and
thread behaviors. We have adopted an automata
based description similar to AADL mode transi-
tions.

2 Behaviour descriptions for

AADL

This proposal1 introduces some HRT-HOOD con-
cepts within AADL and attach behaviours to sub-
programs and threads. HRT-HOOD concepts are
speci�ed through dedicated AADL properties. A be-
havioural annex extending AADL mode automata is
de�ned to specify subprogram and thread behaviours.
In the following, we present the behavioral annex at-
tached to AADL components and how synchroniza-
tion mechanisms existing in HRT-HOOD can be spec-
i�ed in the AADL framework. Lastly, we extend the
behavioral speci�cation to take into account real-time
aspects.

2.1 Subprogram and thread be-

haviour

An AADL mode automaton is de�ned with a set of
modes with an initial mode and mode transitions en-
abled by events, using the syntax:

m1 -[ event ]-> m2

In the proposed annex, we use the same syntax
to de�ne state transitions. However, an action is at-
tached to a transition and activation condition can be
speci�ed by an event, a predicate or both. Further-
more, besides the initial keyword of AADL modes,
we introduce a return keyword for subprogram im-
plementation states. When reached, control returns
to the caller. However, the automaton can be non de-
terministic and may specify several possible e�ective
implementations. Moreover, within an automaton,
we can access the subprogram parameters according

1In fact, this proposal has been elaborated from the one[7]
already done in the context of AADL 0.95.

to their declared mode (in or out). The following
text illustrates the attachment of an automaton to a
subprogram.

subprogram addition

x: in parameter std::integer;

y: in parameter std::integer;

r: out parameter std::integer;

ovf: out parameter std::integer;

end addition;

subprogram implementation addition.default

annex cotre {**

states

s0 : initial state;

s1 : return state;

transitions

s0 -[ ]-> s1 { r := x + y ; ovf := false; }

s0 -[ ]-> s1 { r := 0; ovf := true; }

**};

end addition.default;

Whereas AADL �ows denote sequences of calls, the
proposed behavioural annex allows the expression of
control �ows which take into account state variables
and parameters.

2.1.1 Thread behaviour

The behavior of a thread can be speci�ed using the
same annex. However, the thread terminates if a �nal
state is reached. Termination is not mandatory: the
thread can be active forever.

2.1.2 Subprogram invocation

The AADL standard supports the declaration of sub-
program call sequences which may be conditionned
by a mode. Calls are identi�ed by an occurrence.
The occurrence identi�er is used to specify data �ows
between in and out parameters of called subpro-
grams. For example, consider the addition subpro-
gram which may raise an event if an over�ow occurs:

subprogram addition

features

x: in parameter std::integer;

y: in parameter std::integer;

r: out parameter std::integer;

ovf: out event;

end addition;

The following test subprogram performs two calls
to the addition subprogram. The implementation

of the subprogram contains two calls to addition,
identi�ed by add1 and add2. The connexion section
speci�es how data �ows from local variables to in pa-
rameters of the �rst call and then to in parameters of
the second call:

subprogram test

features

ovf: out event;

end test;

subprogram implementation test.default

subcomponents

x: data Cotre::integer;

y: data Cotre::integer;

z1: data Cotre::integer;

z2: data Cotre::integer;

-- AADL standard:

-- specification of call occurrences

calls {

add1: subprogram addition;

add2: subprogram addition;

};

-- fixed data flow graph

connections

parameter x -> add1.x;

parameter y -> add1.y;

parameter add1.z -> z1;

parameter z1 -> add2.x;

parameter y -> add2.y;

parameter add2.z2 -> z;

event add1.ovf -> ovf;

event add2.ovf -> ovf;

end test.default;

However, such a speci�cation is very restrictive.
The order in which calls are performed is �xed and
cannot be data dependent. Di�erent call sequences
can be associated to di�erent modes. The annex
we propose can specify data dependent behaviors.
The implementation of a subprogram can be speci-
�ed as performing other subprogram calls or event
noti�cations. The same syntax is used for both:
p!x1,...,xn express the call to subprogram p with
actual parameters x1,...xn or the sending of an event
to the port p. If p is a data port or port group, the
values x1,...,xn are transmitted:

annex cotre_behavior {**

states

s0: initial state;

s1: state;

s2: final state;

transitions

s0 -[addition!x,y,z1]-> s1 {}

s1 -[when z1 < 10 &

addition!z1,y,z2]-> s2 {}

s1 -[when z1 >= 10]-> s2 { z2 := z1; }

**};

Remark Operation calls and event noti�cations
can also appear in the action associated with a tran-
sition. In this case, we will write:

annex cotre_behavior {**

states

s0: initial state;

s1: state;

s2: final state;

transitions

s0 -[ ]-> s1 { addition!x,y,z1; }

s1 -[when z1 < 10]-> s2

{ addition!z1,y,z2; }

s1 -[when z1 >= 10]-> s2 { z2 := z1; }

**};

2.1.3 Sendind and receiving messages

Messages are sent and received through AADL ports
or groups of ports according to a syntax which is sim-
ilar to that of subprogram calls. The ! symbol used
to indicate a subprogram call also indicates that a
message has been sent when the identi�er is a port or
group of ports. The parameters are data sent through
the port or through each port in the group. Paren-
theses can be used to separate the sub-groups of a
hierarchical group of ports. The ? symbol indicates
that a message is received from a port or a group of
ports.

thread test

features

p_in: in event data port Cotre::integer;

p_out: out event data port Cotre::integer;

end test;

thread implementation test.default

subcomponents

x: data Cotre::integer;

annex cotre_behavior {**

states

s0: initial state;

s1: state;

transitions

s0 -[p_in?x]-> s1 {}

s1 -[p_out!x+1]-> s0 {}

**};

end test.default;

Remark As for operation calls, sent / received mes-
sages can also appear in the action part, but the se-
mantics are di�erent: a transition is selected from
among those which can be immediately �red. If a
synchronization is in the action part, the transition
waits for the synchronization whereas if the synchro-
nization is in the event part of the transition, the �r-
ing of the transition is conditionned by the selection
of this transition among all the enabled transitions.

2.2 Passive objects

Passive objects are translated directly to AADL data
components. A data declaration corresponds to a
class declaration encapsulating data and exporting
access methods. However, the pro�le of methods
must be declared outside the component via a sub-
program component type. The data component and
the subprograms de�ning the method pro�les can be
grouped together in a package.

subprogram put

features

v: in parameter Cotre::integer;

end put;

subprogram get

features

v: out parameter Cotre::integer;

end get;

subprogram empty

features

v: out parameter Cotre::boolean;

end empty;

subprogram full

features

v: out parameter Cotre::boolean;

end full;

data stack

features

put: subprogram put;

get: subprogram get;

empty: subprogram empty;

full: subprogram full;

end stack;

A behavior can be associated with these subpro-
grams in an annex. However, this behavior would
need to access to the implementation of the data
structure, which would require either adding a new
parameter (a reference to the current instance), ei-
ther an AADL connection between each subprogram
and the data. It seems better to include the subpro-
gram behaviors in the data structure implementation.
For this purpose, the annex must declare several en-
try points. Initial states are attached to each subpro-
gram2.

data implementation stack.default

elems: data Cotre::integer

{ Multiplicity => 10; };

sp: data Cotre::integer;

annex cotre_behavior {**

inits

sp := 0;

states

g : initial state for get;

p : initial state for put;

e : initial state for empty;

f : initial state for full;

s1 : return state;

transitions

g -[when sp > 0]->s1

{ sp := sp - 1; v := elems[sp]; }

p -[when sp < 10]->s1

{ elems[sp] := v; sp := sp + 1; }

e -[]->s1 { v := (sp = 0); }

f -[]->s1 { v := (sp = 10); }

**};

end stack.default;

A single annex can contain the behavior for sev-
eral subprograms via several initial state declarations.
The behavior of several subprograms can also be dis-
tributed in di�erent annexes. The action part of a
transition can call a subprogram of the current data
or of another data.

2.3 Shared objects

AADL supports the notion of protected object: a
protected object is a data component whose Con-
currency_Control_Protocol property is positioned

2We reuse the multiplicity property of UML to specify
arrays.

at protected. The access control mechanisms are
actually left open by AADL and must be de�ned
via the Supported_Concurrency_Control_Protocols
property.
Taking the example of the stack, we will add a

property to the speci�cation:

data stack

features

put: subprogram put;

get: subprogram get;

empty: subprogram empty;

full: subprogram full;

properties

Concurrency_Control_Protocol => protected

end stack;

2.4 Active objects

As in HRT-HOOD, the proposed modelization of ac-
tive objects allows the separation of synchronization
from computation. Active objects specify several en-
try points of which behavior is essentially sequencial,
as described in the previous paragraph. Synchroniza-
tion modes are expressed via events or the AADL
client-server mode. However, these notions are not
su�cient to express activation conditions. To do this,
we attach a behavior to the thread of the active ob-
ject. This behavior is also described by an automa-
ton and expresses the activation conditions. Conse-
quently, one or several annexes of an active object
describe the behavior of the entry points and the be-
havior of the active object itself. The communication
between the clients, the object thread and the entry
points is speci�ed by the synchronization mode at-
tached to the entry point. Three modes taken from
HRT-HOOD are proposed and extend AADL client
server protocol:

• ASER: the invocation is non blocking. These
invocations are stored in a queue which can be
bounded through the attribute Queue_Size.

• HSER: the invocation is blocking. The caller re-
sumes once the call returns.

• LSER: the invocation is blocking. The caller re-
sumes once the call is accepted.

For such purposes, we have introduced an AADL
property declared as follows:

Server_Call_Protocol : type enumeration

(ASER,HSER,LSER) → HSER

applies to (server subprogram);

Remark: HSER is the implicit property attached to
subprograms. Actually, it is the implicit subprogram
call protocol in AADL.

2.4.1 Asynchronous client-server mode

In the asynchronous mode several clients can invoke
the services of a server object. The invocation is non-
blocking. The requested processing is carried out in
parallel with the client. The entry points are rep-
resented by ports which may support data. These
events are stored in �les that can be bounded via the
Queue_Size attribute. The activation conditions are
de�ned by the behavior of the thread associated with
the active object. This thread can accept the events
issued by the clients and may call local subprograms
which perform the requested processing.
For example, the sender thread of the alternated

bit protocol can be speci�ed in AADL extended by
the cotre behavioral annex as follows: the thread
speci�cation declares two input event ports receiv-
ing acknowledgements of each sign, two ouput data
ports transmitting data to the receiver, each port be-
ing associated with a sign, and one input data port
to get information to be transmitted.

thread sender

features

-- data to be transmitted

put: in event data port;

ack0: in event port; -- positive ack

ack1: in event port; -- negative ack

-- data transmission with negative tag

put0: out event data port;

-- data transmission with positive tag

put1: out event data port;

end sender;

The implementation sends data with a given sign
until it receives an acknowledgement of the same sign.
As either data or acknowledgement can be lost, data
must be sent again. A timeout should be speci�ed in
order to avoid to frequent data sendings. When the
acknowledgement has been received a new value is
obtained from the put port and transmitted through
ports associated to the opposite sign.

thread implementation sender.default

v: data;

annex cotre_behavior {**

states

s0: initial state;

s01, s02, s1, s11, s12: state;

transitions

s0 -[put?v]-> s01 {}

s01 -[put0!v]-> s02 {}

s02 -[ack0?]-> s1 {}

s02 -[]-> s01 {} -- timeout

s1 -[put?v]-> s11 {}

s11 -[put1!v]-> s12 {}

s12 -[ack1?]-> s0 {}

s12 -[]-> s11 {} -- timeout

**}

end sender.default;

We propose here to extend the client/server sub-
program feature of AADL so that it supports asyn-
chronous communications. The ASER mode must be
attached to asynchronous entry points. The sender
thread of the alternated bit protocol is then declared
as follows. Input ports become server subprograms
while output ports are declared as required subpro-
grams: they will be connected to entry points of the
communication channel.

thread sender

features

put: server subprogram put

{ Server_Call_Protocol => ASER; };

ack0: server subprogram ack

{ Server_Call_Protocol => ASER; };

ack1: server subprogram ack

{ Server_Call_Protocol => ASER; };

put0: requires subprogram put;

put1: requires subprogram put;

end sender;

The implementation remains unchanged as mes-
sage sending and subprogram calls have the same
syntax. It has to be noted that the subprograms de-
clared here do not need to be implemented. All the
processing is performed by the thread.

2.4.2 Synchronous client-server mode

The HSER mode of HRT-HOOD enables the ser-
vice which has been called up to return a result
to the client. The corresponding synchronization is
the client-server mode of AADL. Remember that the
client-server mode of AADL is not able to take ac-
count of activation conditions.
The proposed solution involves using the object

thread to express the acceptance conditions of the re-
quests sent to each server subprogram. The operation
activation constraints can be expressed via a 3-way
synchronization where the client and the server must

be ready to synchronize and execute the subprogram
code.
In the example below, it is possible to call the put

operation (resp: get) if the bu�er is empty (resp.
full). These conditions are expressed by the server's
behavior automaton. The AADL client-server syn-
chronization is performed after the condition has
been checked.
The semantics of the component behavior is there-

fore as follows: a thread is created and executes
the code from the initial state. The thread is sus-
pended on a pending event which corresponds to an
entry point. Because the client/server communica-
tion mode is synchronous, the client is suspended un-
til completion. Completion occurs when the called
subprogram reaches a return state.

thread serveur

features

put: server subprogram buffer.put;

get: server subprogram buffer.get;

end serveur;

thread implementation serveur.default

subcomponents

buf: data buffer;

annex cotre_behavior {**

states

empty: initial state;

p : initial state for put;

g : initial state for get;

r : return state;

full : state;

transitions

-- expression of acceptance condition

empty -[put?v]-> full { };

full -[get?v]-> empty { };

-- code for entry points

p -[]-> r { buf.put!v; }

g -[]-> r { buf.get!v; }

**}

end serveur.default;

Remarks It may be possible to express the accep-
tance conditions concerning the subprogram param-
eters. The example below only accepts put calls with
a positive argument. A call with a negative or null
argument blocks the client.

annex cotre_behavior {**

states

full : state;

empty : initial state;

transitions

empty -[ put ?x & x > 0 ]-> full { };

full -[ get ?]-> empty { };

**}

2.4.3 Semi-Synchronous client-server mode

In the HRT-HOOD semi-synchronous mode, the
client is unblocked when the message is received. In
this mode, only the input parameters are transmit-
ted. The server thread synchronizes with the mes-
sage sent by the client and calls a local subprogram
which processes the request. The synchronization
(performed by the thread associated with the server)
and the computation (performed by the subprogram)
are always separate.

The semi-synchronous and synchronous modes are
distinguished by the client wakeup time (which oc-
curs when the request is accepted for the semi-
synchronous mode and at the end of processing for
the synchronous mode). To do this, we associate re-
spectiveley the LSER and HSER property values to
the entry point.

The following example declares the semi-
synchronous entry point put. Only positive
data are stored. Attempting to store negative or null
data results in client suspension. The client is woken
up before the e�ective subprogram call by the server
thread.

thread serveur

features

put: server subprogram buffer.put;

{ Server_Call_Protocol => LSER; };

end serveur;

thread implementation serveur.default

subcomponents

buf: data buffer;

annex cotre_behavior {**

states

s0 : initial state;

p: initial state for put;

r: return state;

transitions

-- client wake up if x > 0

s0 -[ put?x & x > 0]-> s0 { };

p -[]-> r { buf.put!x };

**}

end serveur.default;

2.4.4 Combining synchronization modes

The homogeneous declaration of the various synchro-
nization mechanisms makes it easier the speci�cation
of the interface of a thread. The following exam-
ples describes a monitor allocated read or write ac-
cess rights to clients. The entry points start_read
and start_write are synchronous: a client must wait
for the completion of the call before accessing the
ressource. The entry points end_read and end_write
can be asynchronous: the client does not need to wait
for the completion of the request.

thread RW_monitor

features

start_read: server subprogram start_read

{Server_Call_Protocol => HSER; };

end_read: server subprogram end_read

{Server_Call_Protocol => ASER; };

start_write: server subprogram start_write

{Server_Call_Protocol => HSER; };

end_write: server subprogram end_write

{Server_Call_Protocol => ASER; };

end RW_monitor;

2.5 Timing aspects

Timing aspects are already present in AADL through
properties associated to various components. How-
ever, the behavioral extension must also provide im-
plementation related timing constructs.

2.5.1 Elapse of time

Apart from assignments, procedure calls and event
noti�cation, it is possible to associate timed actions
with a transition:

• Computation(min,max) expresses use of the cpu
for a non-deterministic period of time between
min and max.

• Delay(min,max) expresses a suspension for a
non-deterministic period of time between min
and max.

Remark: with respect to the elapse of time, these
two actions express the same thing. However, they
are meaningful with respect to scheduling purposes.

2.5.2 Periodic threads

A periodic thread is declared using pre-declared
AADL attributes: Dispatch_Protocol=>Periodic

and its period Period. The behavior described
by the Cotre annex is triggered from its initial
state and must reach a �nal state before the
Compute_Deadline.

thread Emetteur

features

lput: subprogram put;

end Emetteur;

thread implementation Emetteur.default

properties

Dispatch_Protocol => Periodic;

Period => 10 ms;

annexe cotre.behavior {**

states

s0: initial state;

s1: final state;

transitions

s0 -[ lput ! ]-> s1 { };

**}

end Emetteur.default

2.5.3 Interaction within a bounded time

Several techniques can be used to bound the wait-
ing time for a synchronization when a subprogram
is called, depending on the declaration to which the
attribute specifying a time boundary is attached:

• to subprogram declarations, in accordance with
the HRT-HOOD syntax,

• to subprogram import declarations,

• to subprogram calls or message dispatch.

In order to simplify the semantics of the bounded
wait, a boundary can only be attached to a synchro-
nization on a port or subprogram in LSER mode.
If a HSER call could be bounded, we would have a
problem with respect to the atomicity of the server
computation.
The solution we have chosen to express the

bounded wait involves adding a timeout transition
which is triggered when the system has remained in
a given state for a certain amount of time. Thus, the
time is counted down from the moment the client is
ready to synchronize.
The timeout is introduced via a logic expression

which takes the form g timeout T which is satis�ed
if g has been true for at least T units of time. Given

that the transitions are �red as soon as possible, a
synchronization with timeout is expressed as follows:

s0 -[ g & p! ]-> s1 { }

s0 -[ g & timeout T ]-> s2 { }

3 Properties speci�cation

In the context of real time or concurrent program-
ming, the properties generally address the dynamic
behavior of the system - notions which are harder to
capture using only predicate calculus. These proper-
ties are more often, and more easily, expressed using
temporal logics assertions [5, 1]. In order to help a
user - not necessarily a specialist of temporal logics,
we propose an interface using so called �domain prop-
erties� that we present now. Those �domain proper-
ties� are the sets of generic properties a user working
in this particular �eld is most likely to wish to ver-
ify. To allow speci�cation of more speci�c properties,
we also considered the introduction of �tool proper-
ties� directly written in the speci�cation language of
a given veri�cation tool. In this draft, we will only
discuss properties of the former kind.
The �domain properties� are organized into two

categories:
(1) a set of �general� properties
(2) a set of �component speci�c� properties
For each property, we give an equivalent one in

terms of basic operators.

3.1 General properties

We distinguish two kinds of properties expressing a
general behavior of a speci�c component. Depending
on whether a property is located in the root compo-
nent or in a sub-component, one speci�es a global or
a local property.

• The �rst kind is �state oriented� and speci�es the
possibilities for a component (system) to go back
to an initial state.

• The second kind is �event-oriented� and speci�es
liveness of the component (system) or the pos-
sibility of divergence (in�nite computation with-
out �observable� event).

Re-initializable

• resettable_pot (≡ AG EF initial)

The component may eventually return to its
initial state (from any state).

• resettable_inev (≡ AG AF initial)

The component must eventually return to its
initial state (from any state).

Component liveness Let C be the subset of
events of a component. We extend CTL operators
EF and AF to EFC and AFC . EFC speci�es that
some path from starting from the current state con-
tains a transition of C, AFC speci�es that all paths
starting from the current state contains a transition
of C.

• is_alive (≡ AG EFC true)

Always some action of the component must
be possible in the future

When used in the root component, this property
implies absence of deadlock.

• no_livelock (≡ AG AFC true)

A component may not be inde�nitely idle
(without performing some action)

3.2 Speci�c properties

In order to specify more speci�c properties, we pro-
pose also a small set of �properties patterns�. In the
sequel, e1, e2 . . . denote boolean expression on atomic
variables characterizing the behavior of the compo-
nent.

invariant p (≡ AG p)
Property p always holds

Leadsto

e1 leadsto e2 [within d] ≡ AG (e1 ⇒ AF≤d e2)
When expression e1 holds then expression e2 will

eventually holds [at most in d unit time later].

Reachable from

reachable e1 [from e2][within d]≡ AG(e1 ⇒
EF≤d e2)
When expression e1 holds then expression e2 po-

tentially holds in the future [at most in d unit time
later].

After

e1 after e2 ≡ A[¬e2 Weak Until e1]
When expression e1 holds then expression e2 has

been necessarily satis�ed in the past

Remark: Taking bene�t of the presence of la-
bels of transitions it is also possible to envisage the
use of behavioral equivalences (bisimulation, testing
equivalence, . . . ) to specify directly by means of an
abstract automaton the desired behavior of the con-
sidered part of the system (component system).

4 Conclusion

We have considered behavioural aspects within an ar-
chitecture description language, namely AADL. Our
proposition has tried to keep as much as possible close
to AADL. We are now investigating how we can de-
�ne a formal semantics to some of the AADL con-
structs [3]. Moreover, starting from our experience
in the Cotre project [2], for veri�cation purposes, we
will also investigate the mapping of AADL constructs
to those available within veri�cation tools and more
generally how to deal with AADL transformations [6].

